Maple-assisted proof of formula for A181214

Robert Israel

30 April 2018

There are $2^6 = 64$ possible configurations for a 2×3 sub-array. Consider the 64×64 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×3 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both configurations, and neither the diagonal nor antidiagonal is (1,1,1)), and 0 otherwise. The following Maple code computes it.

```
> Configs:= [seq(convert(x,base,2)[1..6],x=2^6..2^7-1)]:
> T:= Matrix(64,64,(i,j) -> `if`(Configs[i][4..6] = Configs[j][1..3] and Configs[i][1]+Configs[i][5]+Configs[j][6] <= 2 and Configs[i][3]+Configs[i][5]+Configs[j][4] <= 2,1,0)):
```

Thus $a(n) = u\, T^n\, v$ where u and v are row and column vectors respectively with $u_i = 1$ for i corresponding to configurations with bottom row $(0, 0, 0)$, 0 otherwise, and $v_i = 1$ for i corresponding to configurations with top row $(0, 0, 0)$, 0 otherwise. The following Maple code produces these vectors.

```
> u:= Vector[row](64, i -> `if`(Configs[i][1..3] = [0,0,0],1,0)):
  v:= Vector(64, j -> `if`(Configs[j][4..6] = [0,0,0],1,0)):
```

To check, here are the first few entries of our sequence.

```
> seq(u . T^n . v,n=0..10);
```

$$1, 8, 64, 400, 2500, 16100, 103684, 665252, 4268356, 27399292, 175880644 \tag{1}$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \mapsto t^{18} - 7\,t^{17} + 26\,t^{15} - 44\,t^{14} + 216\,t^{13} - 32\,t^{12} - 592\,t^{11} + 315\,t^{10} - 157\,t^9 + 64\,t^8 + 414\,t^7 - 176\,t^6 - 4\,t^5 - 40\,t^3 + 16\,t^2 \tag{2}$$

This turns out to have degree 18. Thus we will have $0 = u\, P(T)\, T^n\, v = \sum_{i=0}^{18} p_i\, a(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 18, which would hold true for any u and v. It seems that with our particular u and v we have a recurrence of order only 9, corresponding to a factor of P.

```
> Q:= unapply(8-12*t-32*t^3-78*t^4+21*t^5+16*t^6+6*t^8-t^9, t);
```

$$Q := t \mapsto -t^9 + 6\,t^8 + 16\,t^6 + 21\,t^5 - 78\,t^4 - 32\,t^3 - 12\,t + 8 \tag{3}$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 9.

```
> R:= unapply(normal(P(t)/Q(t)), t);
```

$$R := t \mapsto -(t^7 - t^6 - 6\,t^5 + 6\,t^4 - 3\,t^3 + 3\,t^2 + 2\,t - 2)\,t^2 \tag{4}$$

Now we want to show that $b(n) = u\, Q(T)\, T^n\, v = 0$ for all n. This will certainly satisfy the order-9 recurrence

$$\sum_{i=0}^{9} r_i\, b(i+n) = \sum_{i=0}^{9} r_i u\, Q(T)\, T^{n+i} v = u\, Q(T)\, R(T)\, T^n\, v = u\, P(T)\, T^n\, v = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(0) = \ldots = b(9) = 0$.

This might take some time to do naively, so it's worthwhile to do some pre-calculation.

```
> w:= u . Q(T):
  V[0]:= v:
  for i from 1 to 9 do V[i]:= T . V[i-1] od:
  seq(w . V[i],i=0..9);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \tag{5}$$